

NAG Library Routine Document

D03NDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D03NDF computes an analytic solution to the Black–Scholes equation for a certain set of option types.

2 Specification

```
SUBROUTINE D03NDF (KOPT, X, S, T, TMAT, TDPAR, R, Q, SIGMA, F, THETA,
& DELTA, GAMMA, LAMBDA, RHO, IFAIL)

INTEGER KOPT, IFAIL
REAL (KIND=nag_wp) X, S, T, TMAT, R(*), Q(*), SIGMA(*), F, THETA, DELTA,
& GAMMA, LAMBDA, RHO
LOGICAL TDPAR(3)
```

3 Description

D03NDF computes an analytic solution to the Black–Scholes equation (see Hull (1989) and Wilmott *et al.* (1995))

$$\frac{\partial f}{\partial t} + (r - q)S \frac{\partial f}{\partial S} + \frac{\sigma^2 S^2}{2} \frac{\partial^2 f}{\partial S^2} = rf \quad (1)$$

$$S_{\min} < S < S_{\max}, \quad t_{\min} < t < t_{\max}, \quad (2)$$

for the value f of a European put or call option, or an American call option with zero dividend q . In equation (1) t is time, S is the stock price, X is the exercise price, r is the risk free interest rate, q is the continuous dividend, and σ is the stock volatility. The parameter r , q and σ may be either constant, or functions of time. In the latter case their average instantaneous values over the remaining life of the option should be provided to D03NDF. An auxiliary routine D03NEF is available to compute such averages from values at a set of discrete times. Equation (1) is subject to different boundary conditions depending on the type of option. For a call option the boundary condition is

$$f(S, t = t_{\text{mat}}) = \max(0, S - X)$$

where t_{mat} is the maturity time of the option. For a put option the equation (1) is subject to

$$f(S, t = t_{\text{mat}}) = \max(0, X - S).$$

D03NDF also returns values of the Greeks

$$\Theta = \frac{\partial f}{\partial t}, \quad \Delta = \frac{\partial f}{\partial x}, \quad \Gamma = \frac{\partial^2 f}{\partial x^2}, \quad \Lambda = \frac{\partial f}{\partial \sigma}, \quad \rho = \frac{\partial f}{\partial r}.$$

S30ABF also computes the European option price given by the Black–Scholes–Merton formula together with a more comprehensive set of sensitivities (Greeks).

Further details of the analytic solution returned are given in Section 8.1.

4 References

Hull J (1989) *Options, Futures and Other Derivative Securities* Prentice–Hall

Wilmott P, Howison S and Dewynne J (1995) *The Mathematics of Financial Derivatives* Cambridge University Press

5 Parameters

1: KOPT – INTEGER *Input*

On entry: specifies the kind of option to be valued:

KOPT = 1

A European call option.

KOPT = 2

An American call option.

KOPT = 3

A European put option.

Constraints:

KOPT = 1, 2 or 3;

if $q \neq 0$, KOPT $\neq 2$.

2: X – REAL (KIND=nag_wp) *Input*

On entry: the exercise price X .

Constraint: $X \geq 0.0$.

3: S – REAL (KIND=nag_wp) *Input*

On entry: the stock price at which the option value and the Greeks should be evaluated.

Constraint: $S \geq 0.0$.

4: T – REAL (KIND=nag_wp) *Input*

On entry: the time at which the option value and the Greeks should be evaluated.

Constraint: $T \geq 0.0$.

5: TMAT – REAL (KIND=nag_wp) *Input*

On entry: the maturity time of the option.

Constraint: $TMAT \geq T$.

6: TDPAR(3) – LOGICAL array *Input*

On entry: specifies whether or not various parameters are time-dependent. More precisely, r is time-dependent if TDPAR(1) = .TRUE. and constant otherwise. Similarly, TDPAR(2) specifies whether q is time-dependent and TDPAR(3) specifies whether σ is time-dependent.

7: R(*) – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array R must be at least 3 if TDPAR(1) = .TRUE., and at least 1 otherwise.

On entry: if TDPAR(1) = .FALSE. then R(1) must contain the constant value of r . The remaining elements need not be set.

If TDPAR(1) = .TRUE. then R(1) must contain the value of r at time T and R(2) must contain its average instantaneous value over the remaining life of the option:

$$\hat{r} = \int_T^{TMAT} r(\zeta) d\zeta.$$

The auxiliary routine D03NEF may be used to construct R from a set of values of r at discrete times.

8: Q(*) – REAL (KIND=nag_wp) array Input

Note: the dimension of the array Q must be at least 3 if TDPAR(2) = .TRUE., and at least 1 otherwise.

On entry: if TDPAR(2) = .FALSE. then Q(1) must contain the constant value of q . The remaining elements need not be set.

If TDPAR(2) = .TRUE. then Q(1) must contain the constant value of q and Q(2) must contain its average instantaneous value over the remaining life of the option:

$$\hat{q} = \int_T^{T_{MAT}} q(\zeta) d\zeta.$$

The auxiliary routine D03NEF may be used to construct Q from a set of values of q at discrete times.

9: SIGMA(*) – REAL (KIND=nag_wp) array Input

Note: the dimension of the array SIGMA must be at least 3 if TDPAR(3) = .TRUE., and at least 1 otherwise.

On entry: if TDPAR(3) = .FALSE. then SIGMA(1) must contain the constant value of σ . The remaining elements need not be set.

If TDPAR(3) = .TRUE. then SIGMA(1) must contain the value of σ at time T, SIGMA(2) the average instantaneous value $\hat{\sigma}$, and SIGMA(3) the second-order average $\bar{\sigma}$, where:

$$\hat{\sigma} = \int_T^{T_{MAT}} \sigma(\zeta) d\zeta,$$

$$\bar{\sigma} = \left(\int_T^{T_{MAT}} \sigma^2(\zeta) d\zeta \right)^{1/2}.$$

The auxiliary routine D03NEF may be used to compute SIGMA from a set of values at discrete times.

Constraints:

- if TDPAR(3) = .FALSE., SIGMA(1) > 0.0;
- if TDPAR(3) = .TRUE., SIGMA(i) > 0.0, for $i = 1, 2, 3$.

10: F – REAL (KIND=nag_wp) Output

On exit: the value f of the option at the stock price S and time T.

11: THETA – REAL (KIND=nag_wp) Output

12: DELTA – REAL (KIND=nag_wp) Output

13: GAMMA – REAL (KIND=nag_wp) Output

14: LAMBDA – REAL (KIND=nag_wp) Output

15: RHO – REAL (KIND=nag_wp) Output

On exit: the values of various Greeks at the stock price S and time T, as follows:

$$\text{THETA} = \Theta = \frac{\partial f}{\partial t}, \quad \text{DELTA} = \Delta = \frac{\partial f}{\partial S}, \quad \text{GAMMA} = \Gamma = \frac{\partial^2 f}{\partial S^2},$$

$$\text{LAMBDA} = \Lambda = \frac{\partial f}{\partial \sigma}, \quad \text{RHO} = \rho = \frac{\partial f}{\partial r}.$$

16: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $\text{IFAIL} = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$\text{IFAIL} = 1$

On entry, $\text{KOPT} < 1$,
or $\text{KOPT} > 3$,
or $\text{KOPT} = 2$ when $q \neq 0$,
or $X < 0.0$,
or $S < 0.0$,
or $T < 0.0$,
or $\text{TMAT} < T$,
or $\text{SIGMA}(1) \leq 0.0$, with $\text{TDPAR}(3) = \text{.FALSE.}$,
or $\text{SIGMA}(i) \leq 0.0$, with $\text{TDPAR}(3) = \text{.TRUE.}$, for some $i = 1, 2$ or 3 .

7 Accuracy

Given accurate values of R , Q and SIGMA no further approximations are made in the evaluation of the Black–Scholes analytic formulae, and the results should therefore be within machine accuracy. The values of R , Q and SIGMA returned from D03NEF are exact for polynomials of degree up to 3.

8 Further Comments

8.1 Algorithmic Details

The Black–Scholes analytic formulae are used to compute the solution. For a European call option these are as follows:

$$f = Se^{-\hat{q}(T-t)} N(d_1) - Xe^{-\hat{r}(T-t)} N(d_2)$$

where

$$d_1 = \frac{\log(S/X) + (\hat{r} - \hat{q} + \bar{\sigma}^2/2)(T-t)}{\bar{\sigma}\sqrt{T-t}},$$

$$d_2 = \frac{\log(S/X) + (\hat{r} - \hat{q} - \bar{\sigma}^2/2)(T-t)}{\bar{\sigma}\sqrt{T-t}} = d_1 - \bar{\sigma}\sqrt{T-t},$$

$N(x)$ is the cumulative Normal distribution function and $N'(x)$ is its derivative

$$N(x) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^x e^{-\zeta^2/2} d\zeta,$$

$$N'(x) = \frac{1}{\sqrt{2\pi}} e^{-x^2/2}.$$

The functions \hat{q} , \hat{r} , $\hat{\sigma}$ and $\bar{\sigma}$ are average values of q , r and σ over the time to maturity:

$$\begin{aligned}\hat{q} &= \frac{1}{T-t} \int_t^T q(\zeta) d\zeta, \\ \hat{r} &= \frac{1}{T-t} \int_t^T r(\zeta) d\zeta, \\ \hat{\sigma} &= \frac{1}{T-t} \int_t^T \sigma(\zeta) d\zeta, \\ \bar{\sigma} &= \left(\frac{1}{T-t} \int_t^T \sigma^2(\zeta) d\zeta \right)^{1/2}.\end{aligned}$$

The Greeks are then calculated as follows:

$$\begin{aligned}\Delta &= \frac{\partial f}{\partial S} = e^{-\hat{q}(T-t)} N(d_1) + \frac{Se^{-\hat{q}(T-t)} N'(d_1) - Xe^{-\hat{r}(T-t)} N'(d_2)}{\bar{\sigma} S \sqrt{T-t}}, \\ \Gamma &= \frac{\partial^2 f}{\partial S^2} = \frac{Se^{-\hat{q}(T-t)} N'(d_1) + Xe^{-\hat{r}(T-t)} N'(d_2)}{\bar{\sigma} S^2 \sqrt{T-t}} + \frac{Se^{-\hat{q}(T-t)} N'(d_1) - Xe^{-\hat{r}(T-t)} N'(d_2)}{\bar{\sigma}^2 S^2 (T-t)}, \\ \Theta &= \frac{\partial f}{\partial t} = rf + (q - r)S\Delta - \frac{\sigma^2 S^2}{2}\Gamma, \\ \Lambda &= \frac{\partial f}{\partial \sigma} = \left(\frac{Xd_1 e^{-\hat{r}(T-t)} N'(d_2) - Sd_2 e^{-\hat{q}(T-t)} N'(d_1)}{\bar{\sigma}^2} \right) \hat{\sigma}, \\ \rho &= \frac{\partial f}{\partial r} = X(T-t)e^{-\hat{r}(T-t)} N(d_2) + \frac{(Se^{-\hat{q}(T-t)} N'(d_1) - Xe^{-\hat{r}(T-t)} N'(d_2)) \sqrt{T-t}}{\bar{\sigma}}.\end{aligned}$$

Note: that Θ is obtained from substitution of other Greeks in the Black–Scholes partial differential equation, rather than differentiation of f . The values of q , r and σ appearing in its definition are the instantaneous values, not the averages. Note also that both the first-order average $\hat{\sigma}$ and the second-order average $\bar{\sigma}$ appear in the expression for Λ . This results from the fact that Λ is the derivative of f with respect to σ , not $\hat{\sigma}$.

For a European put option the equivalent equations are:

$$f = Xe^{-\hat{r}(T-t)}N(-d_2) - Se^{-\hat{q}(T-t)}N(-d_1),$$

$$\Delta = \frac{\partial f}{\partial S} = -e^{-\hat{q}(T-t)}N(-d_1) + \frac{Se^{-\hat{q}(T-t)}N'(-d_1) - Xe^{-\hat{r}(T-t)}N'(-d_2)}{\bar{\sigma}S\sqrt{T-t}},$$

$$\Gamma = \frac{\partial^2 f}{\partial S^2} = \frac{Xe^{-\hat{r}(T-t)}N'(-d_2) + Se^{-\hat{q}(T-t)}N'(-d_1)}{\bar{\sigma}S^2\sqrt{T-t}} + \frac{Xe^{-\hat{r}(T-t)}N''(-d_2) - Se^{-\hat{q}(T-t)}N''(-d_1)}{\bar{\sigma}^2S^2(T-t)},$$

$$\Theta = \frac{\partial f}{\partial t} = rf + (q - r)S\Delta - \frac{\sigma^2 S^2}{2}\Gamma,$$

$$\Lambda = \frac{\partial f}{\partial \sigma} = \left(\frac{Xd_1 e^{-\hat{r}(T-t)}N'(-d_2) - Sd_2 e^{-\hat{q}(T-t)}N'(-d_1)}{\bar{\sigma}^2} \right) \hat{\sigma},$$

$$\rho = \frac{\partial f}{\partial r} = -X(T-t)e^{-\hat{r}(T-t)}N(-d_2) + \frac{(Se^{-\hat{q}(T-t)}N'(-d_1) - Xe^{-\hat{r}(T-t)}N'(-d_2))\sqrt{T-t}}{\hat{\sigma}}.$$

The analytic solution for an American call option with $q = 0$ is identical to that for a European call, since early exercise is never optimal in this case. For all other cases no analytic solution is known.

9 Example

This example solves the Black–Scholes equation for valuation of a 5-month American call option on a non-dividend-paying stock with an exercise price of \$50. The risk-free interest rate is 10% per annum, and the stock volatility is 40% per annum.

The option is valued at a range of times and stock prices.

9.1 Program Text

```
! D03NDF Example Program Text
! Mark 24 Release. NAG Copyright 2012.

Module d03ndfe_mod

! D03NDF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp
! .. Implicit None Statement ..
Implicit None
! .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6
Contains
Subroutine print_greek(ns,nt,tmat,s,t,grname,greek)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In)      :: tmat
Integer, Intent (In)                  :: ns, nt
Character (*), Intent (In)           :: grname
! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In)      :: greek(ns,nt), s(ns), t(nt)
! .. Local Scalars ..
```

```

      Integer                      :: i, j
!
!     .. Intrinsic Procedures ..
Intrinsic                      :: len
!
!     .. Executable Statements ..
Write (nout,*)
Write (nout,*) grname
Write (nout,*)('-',i=1,len(grname))
Write (nout,*) ' Stock Price | Time to Maturity (months)'
Write (nout,99999) '|', (12.0_nag_wp*(tmat-t(i)),i=1,nt)
Write (nout,*) ' -----', ('-----',i=1,nt)
Do i = 1, ns
    Write (nout,99998) s(i), '|', (greek(i,j),j=1,nt)
End Do

Return

99999 Format (16X,A,1X,12(1P,E12.4))
99998 Format (1X,1P,E12.4,3X,A,1X,12(1P,E12.4))
End Subroutine print_greek
End Module d03ndfe_mod

Program d03ndfe

!     D03NDF Example Main Program

!
!     .. Use Statements ..
Use nag_library, Only: d03ndf, nag_wp
Use d03ndfe_mod, Only: nin, nout, print_greek
!
!     .. Implicit None Statement ..
Implicit None
!
!     .. Parameters ..
Logical, Parameter           :: gprnt(5) = .True.
!
!     .. Local Scalars ..
Real (Kind=nag_wp)          :: ds, dt, tmat, x
Integer                      :: i, ifail, j, kopt, ns, nt
!
!     .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: delta(:,:), f(:,:),
                                     gamma(:,:),
                                     lambda(:,:),
                                     rho(:,:),
                                     s(:),
                                     t(:),
                                     theta(:,:)
Real (Kind=nag_wp)          :: q(3), r(3), sigma(3)
Logical                      :: tdpard(3)
!
!     .. Intrinsic Procedures ..
Intrinsic                      :: real
!
!     .. Executable Statements ..
Write (nout,*) 'D03NDF Example Program Results'
Write (nout,*)

!
!     Skip heading in data file
Read (nin,*)
Read (nin,*) ns, nt

Allocate (delta(ns,nt),f(ns,nt),gamma(ns,nt),lambda(ns,nt),rho(ns,nt), &
          s(ns),t(nt),theta(ns,nt))

!
!     Read problem parameters

Read (nin,*) kopt
Read (nin,*) x
Read (nin,*) tmat
Read (nin,*) q(1), r(1), sigma(1)
Read (nin,*) s(1), s(ns)
Read (nin,*) t(1), t(nt)
Read (nin,*) tdpard(1:3)

If (ns<2) Then
    Write (nout,*) 'NS invalid.'
Else If (nt<2) Then
    Write (nout,*) 'NT invalid.'
Else
    ds = (s(ns)-s(1))/real(ns-1,kind=nag_wp)

```

```

dt = (t(nt)-t(1))/real(nt-1,kind=nag_wp)

!      Loop over times
Do j = 1, nt
    t(j) = t(1) + real(j-1,kind=nag_wp)*dt

!      Loop over stock prices
Do i = 1, ns
    s(i) = s(1) + real(i-1,kind=nag_wp)*ds

!      Call Black-Scholes solver
ifail = 0
Call d03ndf(kopt,x,s(i),t(j),tmat,tdpar,r,q,sigma,f(i,j), &
            theta(i,j),delta(i,j),gamma(i,j),lambda(i,j),rho(i,j),ifail)

End Do
End Do

!      Output option values and possibly Greeks.

Call print_greek(ns,nt,tmat,s,t,'Option Values',f)

If (gprnt(1)) Call print_greek(ns,nt,tmat,s,t,'Theta',theta)
If (gprnt(2)) Call print_greek(ns,nt,tmat,s,t,'Delta',delta)
If (gprnt(3)) Call print_greek(ns,nt,tmat,s,t,'Gamma',gamma)
If (gprnt(4)) Call print_greek(ns,nt,tmat,s,t,'Lambda',lambda)
If (gprnt(5)) Call print_greek(ns,nt,tmat,s,t,'Rho',rho)

End If

End Program d03ndfe

```

9.2 Program Data

```

D03NDF Example Program Data
21 4                      : ns, nt
2                      : kopt
50.                     : x
0.4166667                : tmat
0.0 0.1 0.4              : q(1), r(1), sigma(1)
0.0 100.                  : s(1), s(ns)
0.0 0.125                 : t(1), t(nt)
.FALSE. .FALSE. .FALSE.   : tdpar

```

9.3 Program Results

D03NDF Example Program Results

Option Values

Stock Price	Time to Maturity (months)			
	5.0000E+00	4.5000E+00	4.0000E+00	3.5000E+00
0.0000E+00	0.0000E+00	0.0000E+00	0.0000E+00	0.0000E+00
5.0000E+00	4.4491E-19	4.5989E-21	1.5461E-23	1.0478E-26
1.0000E+01	5.5566E-10	5.5129E-11	3.1298E-12	8.0281E-14
1.5000E+01	4.7337E-06	1.2187E-06	2.2774E-07	2.7003E-08
2.0000E+01	7.2236E-04	3.1054E-04	1.1005E-04	2.9678E-05
2.5000E+01	1.6557E-02	9.6610E-03	5.0099E-03	2.2012E-03
3.0000E+01	1.3307E-01	9.4037E-02	6.1869E-02	3.6848E-02
3.5000E+01	5.6631E-01	4.5257E-01	3.4667E-01	2.5053E-01
4.0000E+01	1.6004E+00	1.3850E+00	1.1699E+00	9.5640E-01
4.5000E+01	3.4384E+00	3.1328E+00	2.8168E+00	2.4891E+00
5.0000E+01	6.1165E+00	5.7600E+00	5.3874E+00	4.9960E+00
5.5000E+01	9.5300E+00	9.1645E+00	8.7846E+00	8.3882E+00
6.0000E+01	1.3509E+01	1.3163E+01	1.2808E+01	1.2445E+01
6.5000E+01	1.7883E+01	1.7568E+01	1.7251E+01	1.6932E+01
7.0000E+01	2.2513E+01	2.2230E+01	2.1949E+01	2.1671E+01
7.5000E+01	2.7301E+01	2.7045E+01	2.6792E+01	2.6544E+01

8.0000E+01		3.2182E+01	3.1946E+01	3.1713E+01	3.1485E+01
8.5000E+01		3.7117E+01	3.6894E+01	3.6674E+01	3.6458E+01
9.0000E+01		4.2081E+01	4.1868E+01	4.1656E+01	4.1446E+01
9.5000E+01		4.7062E+01	4.6854E+01	4.6647E+01	4.6441E+01
1.0000E+02		5.2052E+01	5.1847E+01	5.1643E+01	5.1439E+01

Theta

Stock Price		Time to Maturity (months)			
		5.0000E+00	4.5000E+00	4.0000E+00	3.5000E+00
0.0000E+00		0.0000E+00	0.0000E+00	0.0000E+00	0.0000E+00
5.0000E+00		-4.4017E-17	-5.5977E-19	-2.3735E-21	-2.0936E-24
1.0000E+01		-2.7827E-08	-3.3857E-09	-2.4163E-10	-8.0398E-12
1.5000E+01		-1.3953E-04	-4.3864E-05	-1.0258E-05	-1.5706E-06
2.0000E+01		-1.3287E-02	-6.9342E-03	-3.0567E-03	-1.0576E-03
2.5000E+01		-1.9512E-01	-1.3714E-01	-8.7730E-02	-4.9018E-02
3.0000E+01		-1.0161E+00	-8.5596E-01	-6.8695E-01	-5.1395E-01
3.5000E+01		-2.8112E+00	-2.6426E+00	-2.4328E+00	-2.1723E+00
4.0000E+01		-5.1662E+00	-5.1709E+00	-5.1500E+00	-5.0892E+00
4.5000E+01		-7.2196E+00	-7.4540E+00	-7.7180E+00	-8.0183E+00
5.0000E+01		-8.3848E+00	-8.7388E+00	-9.1543E+00	-9.6525E+00
5.5000E+01		-8.6152E+00	-8.9372E+00	-9.3056E+00	-9.7329E+00
6.0000E+01		-8.2058E+00	-8.4077E+00	-8.6186E+00	-8.8343E+00
6.5000E+01		-7.5116E+00	-7.5845E+00	-7.6368E+00	-7.6553E+00
7.0000E+01		-6.7905E+00	-6.7711E+00	-6.7202E+00	-6.6262E+00
7.5000E+01		-6.1758E+00	-6.1099E+00	-6.0160E+00	-5.8893E+00
8.0000E+01		-5.7084E+00	-5.6310E+00	-5.5359E+00	-5.4234E+00
8.5000E+01		-5.3786E+00	-5.3103E+00	-5.2340E+00	-5.1533E+00
9.0000E+01		-5.1582E+00	-5.1071E+00	-5.0551E+00	-5.0062E+00
9.5000E+01		-5.0165E+00	-4.9835E+00	-4.9536E+00	-4.9298E+00
1.0000E+02		-4.9281E+00	-4.9107E+00	-4.8979E+00	-4.8916E+00

Delta

Stock Price		Time to Maturity (months)			
		5.0000E+00	4.5000E+00	4.0000E+00	3.5000E+00
0.0000E+00		0.0000E+00	0.0000E+00	0.0000E+00	0.0000E+00
5.0000E+00		3.1381E-18	3.5969E-20	1.3576E-22	1.0494E-25
1.0000E+01		1.4005E-09	1.5376E-10	9.7805E-12	2.8553E-13
1.5000E+01		6.1418E-06	1.7452E-06	3.6436E-07	4.9030E-08
2.0000E+01		5.6040E-04	2.6494E-04	1.0451E-04	3.1863E-05
2.5000E+01		8.3312E-03	5.3217E-03	3.0570E-03	1.5104E-03
3.0000E+01		4.5711E-02	3.5158E-02	2.5461E-02	1.6934E-02
3.5000E+01		1.3765E-01	1.1889E-01	9.9459E-02	7.9557E-02
4.0000E+01		2.8307E-01	2.6258E-01	2.3996E-01	2.1479E-01
4.5000E+01		4.5320E-01	4.3858E-01	4.2214E-01	4.0335E-01
5.0000E+01		6.1427E-01	6.0856E-01	6.0249E-01	5.9601E-01
5.5000E+01		7.4525E-01	7.4687E-01	7.4937E-01	7.5308E-01
6.0000E+01		8.4052E-01	8.4611E-01	8.5298E-01	8.6148E-01
6.5000E+01		9.0433E-01	9.1096E-01	9.1862E-01	9.2752E-01
7.0000E+01		9.4449E-01	9.5045E-01	9.5699E-01	9.6412E-01
7.5000E+01		9.6862E-01	9.7325E-01	9.7808E-01	9.8300E-01
8.0000E+01		9.8260E-01	9.8589E-01	9.8913E-01	9.9221E-01
8.5000E+01		9.9050E-01	9.9269E-01	9.9473E-01	9.9653E-01
9.0000E+01		9.9487E-01	9.9627E-01	9.9748E-01	9.9848E-01
9.5000E+01		9.9725E-01	9.9811E-01	9.9881E-01	9.9935E-01
1.0000E+02		9.9854E-01	9.9905E-01	9.9945E-01	9.9972E-01

Gamma

Stock Price		Time to Maturity (months)			
		5.0000E+00	4.5000E+00	4.0000E+00	3.5000E+00
0.0000E+00		0.0000E+00	0.0000E+00	0.0000E+00	0.0000E+00
5.0000E+00		2.1246E-17	2.7112E-19	1.1536E-21	1.0211E-24
1.0000E+01		3.3102E-09	4.0468E-10	2.9020E-11	9.7029E-13
1.5000E+01		7.2660E-06	2.2982E-06	5.4080E-07	8.3319E-08
2.0000E+01		3.8245E-04	2.0111E-04	8.9333E-05	3.1153E-05
2.5000E+01		3.5190E-03	2.4960E-03	1.6118E-03	9.0924E-04

3.0000E+01		1.2392E-02	1.0554E-02	8.5660E-03	6.4838E-03
3.5000E+01		2.4348E-02	2.3181E-02	2.1626E-02	1.9580E-02
4.0000E+01		3.2765E-02	3.3274E-02	3.3650E-02	3.3795E-02
4.5000E+01		3.4099E-02	3.5763E-02	3.7655E-02	3.9828E-02
5.0000E+01		2.9625E-02	3.1360E-02	3.3403E-02	3.5860E-02
5.5000E+01		2.2600E-02	2.3743E-02	2.5052E-02	2.6569E-02
6.0000E+01		1.5672E-02	1.6137E-02	1.6603E-02	1.7048E-02
6.5000E+01		1.0123E-02	1.0119E-02	1.0032E-02	9.8216E-03
7.0000E+01		6.1999E-03	5.9720E-03	5.6534E-03	5.2154E-03
7.5000E+01		3.6474E-03	3.3666E-03	3.0215E-03	2.6027E-03
8.0000E+01		2.0815E-03	1.8329E-03	1.5510E-03	1.2387E-03
8.5000E+01		1.1610E-03	9.7196E-04	7.7211E-04	5.6851E-04
9.0000E+01		6.3660E-04	5.0529E-04	3.7553E-04	2.5382E-04
9.5000E+01		3.4468E-04	2.5884E-04	1.7950E-04	1.1099E-04
1.0000E+02		1.8494E-04	1.3118E-04	8.4708E-05	4.7786E-05

Lambda

Stock Price		Time to Maturity (months)		
		5.0000E+00	4.5000E+00	4.0000E+00
0.0000E+00		0.0000E+00	0.0000E+00	0.0000E+00
5.0000E+00		8.8525E-17	1.0167E-18	3.8453E-21
1.0000E+01		5.5171E-08	6.0702E-09	3.8694E-10
1.5000E+01		2.7247E-04	7.7565E-05	1.6224E-05
2.0000E+01		2.5496E-02	1.2066E-02	4.7644E-03
2.5000E+01		3.6656E-01	2.3400E-01	1.3431E-01
3.0000E+01		1.8588E+00	1.42448E+00	1.0279E+00
3.5000E+01		4.9710E+00	4.2595E+00	3.5323E+00
4.0000E+01		8.7374E+00	7.9857E+00	7.1787E+00
4.5000E+01		1.1508E+01	1.0863E+01	1.0167E+01
5.0000E+01		1.2344E+01	1.1760E+01	1.1134E+01
5.5000E+01		1.1394E+01	1.0773E+01	1.0104E+01
6.0000E+01		9.4033E+00	8.7137E+00	7.9693E+00
6.5000E+01		7.1285E+00	6.4127E+00	5.6514E+00
7.0000E+01		5.0632E+00	4.3894E+00	3.6936E+00
7.5000E+01		3.4194E+00	2.8406E+00	2.2661E+00
8.0000E+01		2.2203E+00	1.7596E+00	1.3235E+00
8.5000E+01		1.3981E+00	1.0534E+00	7.4380E-01
9.0000E+01		8.5941E-01	6.1393E-01	4.0558E-01
9.5000E+01		5.1846E-01	3.5040E-01	2.1600E-01
1.0000E+02		3.0824E-01	1.9677E-01	1.1294E-01

Rho

Stock Price		Time to Maturity (months)		
		5.0000E+00	4.5000E+00	4.0000E+00
0.0000E+00		0.0000E+00	0.0000E+00	0.0000E+00
5.0000E+00		6.3524E-18	6.5717E-20	2.2112E-22
1.0000E+01		5.6040E-09	5.5594E-10	3.1558E-11
1.5000E+01		3.6414E-05	9.3595E-06	1.7459E-06
2.0000E+01		4.3690E-03	1.8706E-03	6.6008E-04
2.5000E+01		7.9884E-02	4.6268E-02	2.3805E-02
3.0000E+01		5.1594E-01	3.6026E-01	2.3399E-01
3.5000E+01		1.7715E+00	1.3907E+00	1.0448E+00
4.0000E+01		4.0509E+00	3.4193E+00	2.8095E+00
4.5000E+01		7.0648E+00	6.2263E+00	5.3932E+00
5.0000E+01		1.0249E+01	9.2505E+00	8.2458E+00
5.5000E+01		1.3108E+01	1.1967E+01	1.0810E+01
6.0000E+01		1.5384E+01	1.4101E+01	1.2790E+01
6.5000E+01		1.7041E+01	1.5617E+01	1.4153E+01
7.0000E+01		1.8167E+01	1.6613E+01	1.5013E+01
7.5000E+01		1.8894E+01	1.7231E+01	1.5521E+01
8.0000E+01		1.9344E+01	1.7597E+01	1.5806E+01
8.5000E+01		1.9615E+01	1.7807E+01	1.5959E+01
9.0000E+01		1.9774E+01	1.7924E+01	1.6039E+01
9.5000E+01		1.9865E+01	1.7987E+01	1.6080E+01
1.0000E+02		1.9917E+01	1.8022E+01	1.6101E+01

